

Double and triple integrals

If you remember, we defined double and triple integrals in the same fashion as the integral of a single-variable function over an interval: as limit of Riemann sums. To evaluate these integrals, we often convert them into an iterated integral. Let us start with evaluating iterated integrals using Mathematica.

1. Iterated integrals

We use the command **Integrate**. For example, to compute the integral of the function $f(x, y) = x + y$ over the rectangle $R = [0, 2] \times [0, 1]$, we write

```
Integrate[x + y, {x, 0, 2}, {y, 0, 1}]
```

To compute the integral of f over the region D in [Figure 1](#), there are two ways:

```
Integrate[x+y, {x, 0, 2}, {y, x^2, 2 x}]
Integrate[x+y, {y, 0, 4}, {x, y/2, Sqrt[y]}]
```




Figure 1

To compute the integral

$$\int_0^1 \int_{y^2}^y \int_{y+1}^{2+x+y} (x+z) dz dx dy, \quad (1)$$

we write

```
Integrate[x + z, {y, 0, 1}, {x, y^2, y}, {z, y + 1, 2 + x + y}]
```

The integral (1) is a triple integral over a solid. You may wonder what this solid looks like. Based on the bounds of the iterated integral, this solid (called W) is described as

$$W = \{(x, y, z) : 0 \leq y \leq 1, y^2 \leq x \leq y, y + 1 \leq z \leq 2 + x + y\}.$$

Then use the command (see [Figure 2](#)):

```
RegionPlot3D[0 <= y <= 1 && y^2 <= x <= y && y + 1 <= z <= 2 + x + y,
{x, 0, 1}, {y, 0, 1}, {z, 1, 4}, PlotPoints -> 50, AxesLabel -> Automatic]
```




Figure 2

2. Double and triple integrals

Let us try to evaluate the double integral $\int \int_R (x + y) dA$ where R is the region upper half of the unit disc. This region can be described as

$$R = \{(x, y) : x^2 + y^2 \leq 1, y \geq 0\}.$$

With this description, you can evaluate the integral using Mathematica even without writing the double integral as an iterated integral.

```
(* describe the region R *)
R = ImplicitRegion[x^2 + y^2 <= 1 && y >= 0, {x, y}]
```

```
(* evaluate the integral over R *)
Integrate[x + y, {x, y} \[Element] R]
```

After you type `\[Element]`, Mathematica will switch it to symbol \in (belongs to). Similarly, to integral the function $x + z$ over the solid below the plane $z = 2$ and above the paraboloid $z = x^2 + y^2$ (call it W), you describe the region first, then use the command `integrate`.

```
(* describe the region W *)
W = ImplicitRegion[z <= 2 && z >= x^2 + y^2, {x, y, z}]
```

```
(* evaluate the triple integral over W *)
Integrate[x + z, {x, y, z} \[Element] W]
```